


Solutions to Assignment 9

Question 1

The ASCII code for the character 'U' is hex 55 or binary 01010101. Bits are transmitted from LS to MS bit with a space-level start bit and a mark-level stop bit:



Question 2

The CTS (clear to send) handshaking line is an output on a DCE and an input on a DTE. If CTS is an output on the computer's RS-232 port then that computer must be wired up as a DCE. The DSR (data set ready) handshaking line is also an output on a DCE and an input on a DTE. Thus the likely cause of the problem is that both devices are wired as DCEs. A "null modem" could be used. It would cross-connect the data and some handshaking lines. For example, TxD on one connector would be connected to RxD on the other. This should allow the two devices to communicate.

Question 3

The sequence of bits between the HDLC frame sync sequences (01111110) contains one string of five '1's followed by a '0'. This zero is removed. The remaining bits are:

1011 1010 1011 1111 0111

which is BABF7 in hex.

Question 4

One possible solution is shown below:

```

;
; ELEC 464, Assignment 9
; Output string to printer
; Ed Casas, 96/11/23
;

; standard directives for DOS .com files

code    segment public
        assume  cs:code,ds:code
        org     100h

; constants

; note that both the BUSY and STROBE signals pass
; are buffered through inverters so that 0=HIGH
; and 1=LOW

busybit equ     80H      ; BUSY signal (BUSY* bit)
strblo  equ     1100B   ; SEL=1,INIT*=1,AUTOLF=0,STROBE=1
strbhi  equ     1101B   ; SEL=1,INIT*=1,AUTOLF=0,STROBE=0

start:

; use ES=0 to retrieve LPT1 base I/O address

        mov     ax,0
        mov     es,ax
        mov     ax,es:0408H
        mov     base,ax

; initialize control bits

        mov     dx,base
        add     dx,2
        mov     al,strbhi
        out     dx,al
        call    delay

; print string with name and student number

        mov     bx,offset msg ; bx=offset of string
print1: mov     al,[bx]        ; get character
        call    print         ; print it
        or     al,al          ; test it
        jz     done           ; stop if == zero
        inc    bx             ; next character
        jmp    print1         ; repeat
done:

        int     20h           ; return to DOS

; subroutine to print character in AL to the printer

print:

```

```

; save registers and character to be printed
                                code    ends
                                end      start
    push    ax
    push    dx
    mov     tmp,al

; wait while status port BUSY* bit is 0

    mov     dx,base
    add     dx,1
busy:
    in      al,dx
    and     al,busybit
    jz     busy

; put data on interface data pins

    mov     dx,base
    mov     al,tmp
    out     dx,al

; wait setup time, set strobe low, wait minimum
; pulse width, set strobe high and wait hold time

    mov     dx,base
    add     dx,2

    call    delay

    mov     al,strblo
    out     dx,al

    call    delay

    mov     al,strbhi
    out     dx,al

    call    delay

; restore registers & return

    pop     dx
    pop     ax
    ret

; Subroutine to delay for about 1 microsecond.
; Delays are generated by reading the printer
; status port. This assumes that the port is on
; the ISA bus so that the I/O read cycle required
; by an IN instruction will take 4 clock cycles at
; a maximum bus clock frequency of 8 MHz.

delay:
    push    ax
    push    dx
    mov     dx,base
    add     dx,1
    in      al,dx
    in      al,dx
    pop     dx
    pop     ax
    ret

; the message to print (12 is 'form feed')

msg     db     'Ed Casas, 12345678',13,10,12,0
base    dw     ?
tmp     db     ?

```